

NAG Library Routine Document

F07UJF (DTPTRI)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F07UJF (DTPTRI) computes the inverse of a real triangular matrix, using packed storage.

2 Specification

```
SUBROUTINE F07UJF (UPLO, DIAG, N, AP, INFO)
INTEGER          N, INFO
REAL (KIND=nag_wp) AP(*)
CHARACTER(1)     UPLO, DIAG
```

The routine may be called by its LAPACK name *dpttri*.

3 Description

F07UJF (DTPTRI) forms the inverse of a real triangular matrix A , using packed storage. Note that the inverse of an upper (lower) triangular matrix is also upper (lower) triangular.

4 References

Du Croz J J and Higham N J (1992) Stability of methods for matrix inversion *IMA J. Numer. Anal.* **12** 1–19

5 Arguments

- | | |
|--|--------------|
| 1: UPLO – CHARACTER(1) | <i>Input</i> |
| <p><i>On entry:</i> specifies whether A is upper or lower triangular.</p> <p>UPLO = 'U'
 A is upper triangular.</p> <p>UPLO = 'L'
 A is lower triangular.</p> <p><i>Constraint:</i> UPLO = 'U' or 'L'.</p> | |
| 2: DIAG – CHARACTER(1) | <i>Input</i> |
| <p><i>On entry:</i> indicates whether A is a nonunit or unit triangular matrix.</p> <p>DIAG = 'N'
 A is a nonunit triangular matrix.</p> <p>DIAG = 'U'
 A is a unit triangular matrix; the diagonal elements are not referenced and are assumed to be 1.</p> <p><i>Constraint:</i> DIAG = 'N' or 'U'.</p> | |
| 3: N – INTEGER | <i>Input</i> |
| <p><i>On entry:</i> n, the order of the matrix A.</p> <p><i>Constraint:</i> $N \geq 0$.</p> | |

4: AP(*) – REAL (KIND=nag_wp) array *Input/Output*

Note: the dimension of the array AP must be at least $\max(1, N \times (N + 1)/2)$.

On entry: the n by n triangular matrix A , packed by columns.

More precisely,

if $\text{UPLO} = \text{'U'}$, the upper triangle of A must be stored with element A_{ij} in $\text{AP}(i + j(j - 1)/2)$ for $i \leq j$;

if $\text{UPLO} = \text{'L'}$, the lower triangle of A must be stored with element A_{ij} in $\text{AP}(i + (2n - j)(j - 1)/2)$ for $i \geq j$.

If $\text{DIAG} = \text{'U'}$, the diagonal elements of A are assumed to be 1, and are not referenced; the same storage scheme is used whether $\text{DIAG} = \text{'N'}$ or 'U' .

On exit: A is overwritten by A^{-1} , using the same storage format as described above.

5: INFO – INTEGER *Output*

On exit: $\text{INFO} = 0$ unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

$\text{INFO} < 0$

If $\text{INFO} = -i$, argument i had an illegal value. An explanatory message is output, and execution of the program is terminated.

$\text{INFO} > 0$

Element $\langle \text{value} \rangle$ of the diagonal is exactly zero. A is singular its inverse cannot be computed.

7 Accuracy

The computed inverse X satisfies

$$|XA - I| \leq c(n)\epsilon|X||A|,$$

where $c(n)$ is a modest linear function of n , and ϵ is the **machine precision**.

Note that a similar bound for $|AX - I|$ cannot be guaranteed, although it is almost always satisfied.

The computed inverse satisfies the forward error bound

$$|X - A^{-1}| \leq c(n)\epsilon|A^{-1}||A||X|.$$

See Du Croz and Higham (1992).

8 Parallelism and Performance

F07UJF (DTPTRI) makes calls to BLAS and/or LAPACK routines, which may be threaded within the vendor library used by this implementation. Consult the documentation for the vendor library for further information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the OpenMP environment used within this routine. Please also consult the Users' Note for your implementation for any additional implementation-specific information.

9 Further Comments

The total number of floating-point operations is approximately $\frac{1}{3}n^3$.

The complex analogue of this routine is F07UWF (ZTPTRI).

10 Example

This example computes the inverse of the matrix A , where

$$A = \begin{pmatrix} 4.30 & 0.00 & 0.00 & 0.00 \\ -3.96 & -4.87 & 0.00 & 0.00 \\ 0.40 & 0.31 & -8.02 & 0.00 \\ -0.27 & 0.07 & -5.95 & 0.12 \end{pmatrix},$$

using packed storage.

10.1 Program Text

```
Program f07ujfe

!     F07UJF Example Program Text

!     Mark 26 Release. NAG Copyright 2016.

!     .. Use Statements ..
Use nag_library, Only: dtptri, nag_wp, x04ccf
!     .. Implicit None Statement ..
Implicit None
!     .. Parameters ..
Integer, Parameter :: nin = 5, nout = 6
Character (1), Parameter :: diag = 'N'
!     .. Local Scalars ..
Integer :: i, ifail, info, j, n
Character (1) :: uplo
!     .. Local Arrays ..
Real (Kind=nag_wp), Allocatable :: ap(:)
!     .. Executable Statements ..
Write (nout,*) 'F07UJF Example Program Results'
!     Skip heading in data file
Read (nin,*)
Read (nin,*) n

Allocate (ap(n*(n+1)/2))

!     Read A from data file

Read (nin,*) uplo
If (uplo=='U') Then
    Read (nin,*)((ap(i+j*(j-1)/2),j=i,n),i=1,n)
Else If (uplo=='L') Then
    Read (nin,*)((ap(i+(2*n-j)*(j-1)/2),j=1,i),i=1,n)
End If

!     Compute inverse of A
!     The NAG name equivalent of dtptri is f07ujf
Call dtptri(uplo,diag,n,ap,info)

!     Print inverse

Write (nout,*)
Flush (nout)

!     ifail: behaviour on error exit
!           =0 for hard exit, =1 for quiet-soft, =-1 for noisy-soft
ifail = 0
Call x04ccf(uplo,diag,n,ap,'Inverse',ifail)

End Program f07ujfe
```

10.2 Program Data

```
F07UJF Example Program Data
 4 :Value of N
 'L' :Value of UPLO
 4.30
-3.96 -4.87
 0.40  0.31 -8.02
-0.27  0.07 -5.95  0.12 :End of matrix A
```

10.3 Program Results

```
F07UJF Example Program Results
```

Inverse	1	2	3	4
1	0.2326			
2	-0.1891	-0.2053		
3	0.0043	-0.0079	-0.1247	
4	0.8463	-0.2738	-6.1825	8.3333
